AD-A079 529 AEROSPACE CORP EL SEGUNDO CA F/G 20/5

AUTOMATING THE E-BEAM LASER LABORATORY.(U)

NOV 79 R G DE BIASE F04701-79-C-0080

UNCLASSIFIED TR-0080(5940-02)-3 SD-TR-79-10 NL

1 OF 1

AD A079529

END

DATE FILMED

2-80

DDC



MICROCOPY RESOLUTION TEST CHART
NATIONAL BUREAU OF STANDARDS-1963-A

REPORT SD-TR-79-10

LEVEL II

Automating the E-Beam Laser Laboratory

Prepared by R. G. DE BIASE
~~Threat Analysis Office~~
~~Development Group~~
El Segundo, Calif. 90245

30 November 1979

Interim Report

APPROVED FOR PUBLIC RELEASE;
DISTRIBUTION UNLIMITED

Prepared for
SPACE DIVISION
AIR FORCE SYSTEMS COMMAND
Los Angeles, Air Force Station
P.O. Box 92960, Worldway Postal Center
Los Angeles, Calif. 90009

This interim report was submitted by The Aerospace Corporation, El Segundo, CA 90245, under Contract No. F04701-79-C-0080 with the Air Force Space Division, Contracts Management Office, P.O. Box 92960, Worldway Postal Center, Los Angeles, CA 90009. It was reviewed and approved for The Aerospace Corporation by R. D. Turnacliff, Principle Director, Threat Analysis Office, and W. R. Warren, Jr., Director, Aerophysics Laboratory. Gerhard E. Aichinger was the project officer for Mission-Oriented Investigation and Experimentation (MOIE) Programs.

This report has been reviewed by the Information Office (OI) and is releasable to the National Technical Information Service (NTIS). At NTIS, it will be available to the general public, including foreign nations.

This technical report has been reviewed and is approved for publication. Publication of this report does not constitute Air Force approval of the report's findings or conclusions. It is published only for the exchange and stimulation of ideas.

Gerhard E. Aichinger
Project Officer

FOR THE COMMANDER

Frank J. Bane, Chief
Contracts Management Office

UNCLASSIFIED

SECURITY CLASSIFICATION OF THIS PAGE (When Data Entered)

REPORT DOCUMENTATION PAGE

READ INSTRUCTIONS BEFORE COMPLETING FORM

1. REPORT NUMBER
SD-TR-79-10

2. GOVT ACCESSION NO.

3. RECIPIENT'S CATALOG NUMBER

4. TITLE (and Subtitle)
AUTOMATING THE E-BEAM LASER LABORATORY

5. TYPE OF REPORT & PERIOD COVERED
Interim

6. PERFORMING ORG. REPORT NUMBER
TR-0080(5940-02)-3

7. AUTHOR(s)
Raymond G. De Biase

8. CONTRACT OR GRANT NUMBER(s)
F04701-79-C-0080

9. PERFORMING ORGANIZATION NAME AND ADDRESS
The Aerospace Corporation
El Segundo, Calif. 90245

10. PROGRAM ELEMENT, PROJECT, TASK AREA & WORK UNIT NUMBERS

11. CONTROLLING OFFICE NAME AND ADDRESS
Space Division
Air Force Systems Command
Los Angeles Air Aorce Station
Los Angeles, Calif. 90009

12. REPORT DATE
30 Nov[redacted]79

13. NUMBER OF PAGES
27

14. MONITORING AGENCY NAME & ADDRESS (if different from Controlling Office)

15. SECURITY CLASS. (of this report)
Unclassified

15a. DECLASSIFICATION/DOWNGRADING SCHEDULE

16. DISTRIBUTION STATEMENT (of this Report)
Approved for public release; distribution unlimited.

17. DISTRIBUTION STATEMENT (of the abstract entered in Block 20, if different from Report)

18. SUPPLEMENTARY NOTES

19. KEY WORDS (Continue on reverse side if necessary and identify by block number)
Automation
BASIC Program
Data Acquisition
Displays
Laser
Microcomputer
Pulsed Chemical Laser
Sensors

20. ABSTRACT (Continue on reverse side if necessary and identify by block number)
Experiments conducted in the E-beam Laser Laboratory require the setting of gas flows, pressures, and voltages in a timed sequence, with a variety of sensors monitoring the system. The report describes the use of a CROMEMCO microcomputer and other hardware, and details a BASIC program that senses and displays process variables digitally, reads in process parameters interactively, acquires data during an experiment, and computes and prints out test conditions and results.

DD FORM 1473 (FACSIMILE)

UNCLASSIFIED

SECURITY CLASSIFICATION OF THIS PAGE (When Data Entered)

CONTENTS

FIGURES

I. INTRODUCTION

Experiments conducted in the E-beam Laser Laboratory require the setting of gas flows, pressures, and voltages in a timed sequence, to avoid gas mixture explosion boundaries and radiation hazards to personnel. The parameters of an experiment and the variation of the process variables during an experiment impose a burden on the lab operators to read and record these data while setting control valves and voltages to run the test.

A four-phase plan to automate the laboratory is contemplated:

I. Data acquisition from sensors, with a computer summary of the experiment

II. Sampling wideband waveforms and recording data for computer analyses

III. Servo control of valves and optical alignments for automatic operation of the experiment

IV. Data communication to a central computer for analyses of E-beam test results

This report describes the realization of Phase I, which had four objectives:

a. To sense and display the process variables digitally

b. To program a computer to input process parameters interactively

c. To program a computer to acquire data during an experiment

d. To program a computer to print out test conditions and results

Completion of Phase I obviates manual recording of test conditions or results. It should aid in maintaining good test records, and it should increase the rate at which experiments can be conducted in the lab.

II. E-BEAM LASER LABORATORY

Figure 1 portrays, in block diagram form, the interaction among the laboratory's functional components. Operation of the E-beam laser requires metering five gases into the laser chamber. Two protective barricades are employed to hold mixtures of fluorine, oxygen, and helium. Hydrogen and argon are then mixed with the barricade gases in a valve plenum and transported to the laser chamber. The barricade used to initiate filling of the laser chamber contains an oxygen-rich mixture to prevent a low-pressure reaction between fluorine and hydrogen. At a specified laser chamber pressure, a switchover to the oxygen-lean barricade continues charging the chamber. When an appropriate laser chamber pressure and gas mixture is attained, an energetic flux of electrons triggers a chain reaction between fluorine and hydrogen that causes lasing. The energetic electrons are generated by the Pulserad accelerator system.

A dump tank provides a pressure relief volume for the explosive reaction that occurs in the lasing process. It protects the complex cathode structure in the electron gun from damage following a laser shot.

Figure 2 illustrates the two physically separated control and monitoring areas. The "hood" area is adjacent to the laser chamber. Control valves and displays are used to manually set conditions in the laser chamber for tests. When appropriate conditions are set for an experiment, all personnel leave the hood area for the protected "console" area, where the system variables are monitored by digital displays and a Cromemco Z-2D microprocessor. The Pulserad charging system is set to a suitable accelerating voltage, and depression of a "fire" button causes the Marx generator to erect and energetic electrons to irradiate the gas mixture in the laser chamber. As the gases lase, there is an attendant electromagnetic pulse, which induces noise voltages in the sensor system.



Fig. 1. E-Beam Laser Laboratory



Fig. 2. Control and Monitoring

Fourteen analog sensors monitor the setup of an experiment:

a. Three Hastings mass flowmeters with an output signal ranging from 0 to 5 V

b. Five Consolidated Controls pressure transducers with an output range from 0 to 30 mV

c. Three National Semiconductor pressure transducers with an output range from 2.5 to 12.5 V

d. Two Hadron energy meters with an output range from 0 to 10 mV

e. One Omega temperature transducer with an output range from 0 to 2 V

The high noise environment and the varied sensor outputs require signal conditioning by the network, which converts sensor outputs to acceptable computer inputs.

III. DATA CONVERSION NETWORK

The high noise environment dictated that only analog signals would link the hood and console areas. Due to space limitations, sensor power supplies had to be located in the console area. Signal lines from the sensors were shielded from power excitation lines to the sensors. The low bandwidth of the signals between the hood and the console permitted the extensive use of capacitors to reduce high frequency noise pulses. Ground-strapping was employed to minimize ground currents coupling unwanted signals into the operational amplifiers that scale variables for input to the microcomputer.

A scaling and display system for the sensors is shown in Fig. 3. Sensors output analog voltages proportional to physical variables, which must be digitized. Digitization of the variables can be effected with an analog-to-digital (A/D) converter with B converter bits (the dynamic range of the result is 2^B counts). Design constraints specify a maximum voltage and a quantization voltage level for the physical device; therefore, to employ the full



Fig. 3. Data Conversion System

capability of the converter, one must transform the maximum output of the sensor to the design maximum voltage of the converter. Operational amplifier networks provide the voltage adjustments. The Datel digital displays convert analog voltages, in the range of 0 to 1.999 V, to a digital count. To represent a physical variable, e.g., a pressure of 14.7 psia, the input to the display channel is adjusted to 1.470 V. Grounding a selected pin places the decimal point of the display at a chosen position; thus, 14.70 is achieved by grounding the pin for the second decimal point position and represents the pressure (in psia). Each displayed variable is realized in an analogous manner.

IV. COMPUTER PROGRAMMING CONSIDERATIONS

Program development to meet the objectives of Phase I was strongly influenced by a requirement to make it "user-oriented" for E-beam laboratory personnel. It was desired that it be easily modified as alternative operating conditions for tests became necessary. The BASIC programming language meets the requirement for ease of comprehension and ready modification. BASIC is an ideal language for interactively inputing process parameters and printing out test conditions and results. The creation of timing loops for data acquisition from the process sensors in BASIC is not readily achieved. Fortunately, however, the slow dynamic process allowed the use of the Cromemco BASIC's SET and ON ERROR instructions to generate real-time loops.

Figure 4 outlines the BASIC language program "LASER," which was written to monitor operation of the E-beam Laboratory. A listing of the program, variable definitions, and port identifications is given in the Appendix.

Figure 5 illustrates the interactive dialog between operator and computer to input test parameters. Test conditions of the previous experiment are input from a disc file and modified as necessary. To illustrate, a "y" response to the first question accepts the Jan. 5, 1979, date as valid for the current test. An "n" response to the test time question permits the test time for the current experiment to be changed. If an "o" response is given to a question at any point in the interactive dialog, the input is assumed to be complete, and the current test conditions are written onto a disc file.

The data acquisition routine is as follows:

a. Initiated by flow transient from first barricade

b. At one-second intervals:

   1. Reads flowmeters and pressures
   2. Monitors mixer valve temperature
   3. Monitors dump tank pressure



Fig. 4. BASIC Language Program "LASER"

```
 RUN
TEST DATE IS JAN 5, 1979 OK ?
? y
TEST DATE IS JAN 5, 1979
TEST TIME = 900 OK ?
? n
NEW TIME 1300
TEST TIME = 1300 OK ?
? y
TEST TIME = 1300
BARRICADE I FILLED TO 258.5 TORR OXYGEN, 1809.5 TORR OXYGEN
 PLUS FLUORINE, AND 100 PSIA OXYGEN PLUS FLUORINE PLUS HELIUM - OK ?
? n
OXYGEN 245
02 PLUS F2 1600
02 PLUS F2 PLUS HE 147
BARRICADE I FILLED TO 245 TORR OXYGEN, 1600 TORR OXYGEN
 PLUS FLUORINE, AND 147 PSIA OXYGEN PLUS FLUORINE PLUS HELIUM - OK ?
? o
PARAMETER SET-UP COMPLETE
READY TO BEGIN TEST ???
?
```

Fig. 5. Interactive Dialogue

c. Switches to second barricade

d. Pressure transient initiates new sampling interval:

   1. Reads flowmeters and pressures
   2. Monitors mixer valve temperature
   3. Monitors dump tank pressure

e. Fire button

   1. Terminates sampling
   2. Calculates gas composition
   3. Acquires data from 9- and 4-cm calorimeters

Figure 6 illustrates the resulting output print format.

```
E-BEAM LASER                      SHOT NO. 9002                     JAN. 2,1979
TEST TIME 900
                                      PAGE 1
*******************************************************************************

E-BEAM PARAMETERS
 SWITCH PRESSURES                                    DIODE AREA = 10 BY 100 CM
 TRIGATRON = 37.4586 PSIA N2                         A-K GAP = 2 CM
 MARX = 26.8875 PSIA SF6                             DIODE PRESSURE = 2E-04 TORR
 PREPULSE = 72.0332 PSIA SF6                         CHARGING VOLTAGE = 70 KV
 MASTER SWITCH GAP = 2 CM
 WATER RESISTIVITY = 1E+06 OHM-CM

                                    SCOPE RECORDS

                PULSE CHARGE   **************************
                               *                        *
                VERTICAL       *                        *     HORIZ
                               *                        *
                               *                        *
                 250 KV/DIV    *                        *  200 NS/DIV
                               *                        *
                               *                        *
                               **************************


                TUBE VOLTAGE   **************************
                               *                        *
                               *                        *
                VERTICAL       *                        *     HORIZ
                               *                        *
                               *                        *
                 160 KV/DIV    *                        *  50 NS/DIV
                               *                        *
                               *                        *
                               **************************


                ROGOWSKI       **************************
                               *                        *
                               *                        *
                VERTICAL       *                        *     HORIZ
                               *                        *
                               *                        *
                 13 KV/DIV     *                        *  200 NS/DIV
                               *                        *
                               *                        *
                               **************************
```

Fig. 6. Output Print Format

```
E-BEAM LASER                        SHOT NO. 2751                      JAN 5,1979
TEST TIME 1300
                                        PAGE 2

*************************************************************************************

FOIL INSTALLED NOV. 17,1977 SHOTS WITH FOIL 35

GAS HANDLING

FLUORINE BARRICADE I STORAGE VOLUME IN USE = 10 GALS (37.85 LITERS)
BARRICADE I FILLED - DEC. 18,1978 TO 60 TORR OXYGEN, 760 TORR OXYGEN
 PLUS FLUORINE, AND 147 PSIA OXYGEN PLUS FLUORINE PLUS HELIUM
BARRICADE I COMPOSITION - 0.789484 % O2 9.21064 % F2 89.9998 % HE
BARRICADE I PRESSURES -
BEGINNING = 128.2 PSIA          END = 110.9 PSIA
PRESSURE DROP = 17.3 PSI        ELAPSED TIME = 310 SECS
DP/DT = 0.0558064 PSI/SEC       F2 MIX FLOWRATE = 143.691 CC/SEC
THERE ARE 4 SHOTS LIKE THIS LEFT IN BARRICADE I

FLUORINE BARRICADE II STORAGE VOLUME IN USE = 10 GALS (37.85 LITERS)
BARRICADE II FILLED - DEC. 18,1978 TO 6 TORR OXYGEN, 760 TORR OXYGEN
 PLUS FLUORINE, AND 147 PSIA OXYGEN PLUS FLUORINE PLUS HELIUM
BARRICADE II COMPOSITION - 0.0789484 % O2 9.92118 % F2 89.9998 % HE
BARRICADE II PRESSURES -
BEGINNING = 130.6 PSIA          END = 113.1 PSIA
PRESSURE DROP = 17.5 PSI        ELAPSED TIME = 310 SECS
DP/DT = 0.0564516 PSI/SEC       F2 MIX FLOWRATE = 145.352 CC/SEC
THERE ARE 4 SHOTS LIKE THIS LEFT IN BARRICADE II

FLOW RATES                      PRESSURES

 F2 MIX = 140.4 CC/SEC          LASER CHAMBER = 801.5 TORR
 H2     = 7 CC/SEC
 AR     = 84 CC/SEC

MIXTURE COMPOSITION

 GAS              PERCENT               ATM                  TORR

 ARGON            36.36                 0.383453             291.424
 HELIUM           54.55                 0.575284             437.215
 FLUORINE         5.57                  0.0587412            44.6433
 HYDROGEN         3.03                  0.0319544            24.2853
 OXYGEN           0.486                 5.12536E-03          3.89527
```

Fig. 6. Output Print Format (Continued)

E-BEAM LASER TEST TIME 1400 SHOT NO. 2738 JAN 5,1979

PAGE 3

****

LASER OUTPUT

TIME (SECS)	9 CM CAL. (JOULES)	TIME (SECS)	4 CM CAL. (JOULES)
-0.5	-128	-1.5	-128
0	-128	0	-128
0.5	-128	1.5	-128
1	-128	3	-128
1.5	-128	4.5	-128
2	-128	6	-128
2.5	-128	7.5	-128
3	-128	9	-128
3.5	-128	10.5	-128
4	-128	12	-128
4.5	-128	13.5	-128
5	-128	15	-128
5.5	-128	16.5	-128
6	-128	18	-128
6.5	-128	19.5	-128
7	-128	21	-128
7.5	-128	22.5	-128
8	-128	24	-128
8.5	-128	25.5	-128
9	-128	27	-128
9.5	-128	28.5	-128
10	-128	30	-128
10.5	-128	31.5	-128
11	-128	33	-128
11.5	-128	34.5	-128
12	-128	36	-128
12.5	-128	37.5	-128
13	-128	39	-128
13.5	-128	40.5	-128
14	-128	42	-128
14.5	-128	43.5	-128
15	-128	45	-128
15.5	-128	46.5	-128
16	-128	48	-128
16.5	-128	49.5	-128
17	-128	51	-128
17.5	-128	52.5	-128
18	-128	54	-128
18.5	-128	55.5	-128
19	-128	57	-128
19.5	-128	58.5	-128

Fig. 6. Output Print Format (Continued)

V. CROMEMCO Z-2D MICROCOMPUTER

The Z-2D machine employs the Z-80 microprocessor unit and has two 5-1/4-in. minidisc drives. It is an S-100 bus system. A 48k memory was required to accommodate the disc operating system, the 16k Cromemco BASIC, and the "LASER" program. Ninety-four percent of the 48k memory was utilized for the Phase I automation of the E-beam Laboratory.

Auxiliary S-100 boards were used for input/output (I/O). A 4FDC disc controller board provided an interface between the disc and computer memories and also linked an Anderson-Jacobson 841 selectric terminal to the Z-2D bus for console I/O. A TU-ART board with two serial and two parallel I/O channels was used; it contains ten interval timers. The A/D conversion process was effected with the D + 7A board. The system has three boards allowing 21 channels of A/D or D/A conversion. Calibration of the D + 7A boards remained fixed over several months of tests and checks.

The hardware costs are estimated to be:

a.	Cromemco computer:		
	1.	Two 5-1/4-in. floppy discs	
	2.	48k-byte memory	$5,100
b.	A/D converter boards		750
c.	Anderson-Jacobson selectric terminal		1,100
d.	12 digital display channels		2,500
			$9,450

APPENDIX

A listing of the "LASER" program is given on the pages that follow. Variables used in the program are defined in Table A-1. The linear relationship between a physical variable and the counts read at a computer port is given by, e.g.,

$$P8 = CO * XO + XO(O)$$

where P8 represents the laser chamber pressure in torr, XO the counts read at port 11H, CO the scale factor, and XO(O) the offset.

Table A-2 provides the port assignments and function definitions to relate counts to physical variables. Note that the TU-ART switches are the parallel A port (D4 H) and the parallel B port (74 H).

```
LIST

  5 OUT%0074%,0
 10 SFMODE : X=0.12345678 : IF X=0.12345678 THEN RUN
 15 DIM Z$(15),W$(15),Y1$(15),Y2$(15)
 20 DIM V2(2),Q1(2),Q2(2),Q3(2)
 25 OPEN¢1¢"TEMP1"
 30 GET¢1¢N1,H,L,H1,H2,P4,V1,R1,S1,S2,S3,S4,S5,S6,T,N3,R4,R5,R6
 35 OPEN¢2¢"TEMP2"
 40 GET¢2¢V2(1),V2(2),Q1(1),Q1(2),Q2(1),Q2(2),Q3(1),Q3(2)
 45 OPEN¢3¢"TEMP3"
 50 GET¢3¢Z$(-1),W$(-1),Y1$(-1),Y2$(-1)
 55 CLOSE
 60 SET 5,30000.0
 65 @"TEST DATE IS ";Z$;" OK ?"
 70 INPUT D$
 75 IF D$="o"THEN 860
 80 IF D$="y"THEN 95
 85 INPUT"NEW DATE ",Z$
 90 GOTO 65
 95 @
100 @"TEST DATE IS ";Z$
105 @
110 @"TEST TIME = ";T;" OK ?"
115 INPUT D$
120 IF D$="o"THEN 860
125 IF D$="y"THEN 140
130 INPUT"NEW TIME ",T
135 GOTO 110
140 @
145 @"TEST TIME = ";T
150 @
155 @"BARRICADE I FILLED TO ";Q1(1);" TORR OXYGEN, ";Q2(1);" TORR OXYGEN"
160 @" PLUS FLUORINE, AND ";Q3(1);" PSIA OXYGEN PLUS FLUORINE PLUS HELIUM - OK ?"
165 INPUT D$
170 IF D$="o"THEN 860
175 IF D$="y"THEN 200
180 INPUT"OXYGEN ",Q1(1)
185 INPUT"O2 PLUS F2 ",Q2(1)
190 INPUT"O2 PLUS F2 PLUS HE ",Q3(1)
195 GOTO 155
200 @
205 @"BARRICADE I FILLED TO ";Q1(1);" TORR O2, ";Q2(1);" TORR O2 PLUS F2"
210 @" AND ";Q3(1);" PSIA O2 PLUS F2 PLUS HE"
215 @
220 @"BARRICADE II FILLED TO ";Q1(2);" TORR OXYGEN, ";Q2(2);" TORR OXYGEN"
225 @" PLUS FLUORINE, AND ";Q3(2);" PSIA OXYGEN PLUS FLUORINE PLUS HELIUM - OK ?"
230 INPUT D$
235 IF D$="o"THEN 860
240 IF D$="y"THEN 265
245 INPUT"OXYGEN ",Q1(2)
250 INPUT"O2 PLUS F2 ",Q2(2)
255 INPUT"O2 PLUS F2 PLUS HE ",Q3(2)
260 GOTO 220
265 @
270 @"BARRICADE II FILLED TO ";Q1(2);" TORR O2, ";Q2(2);" TORR O2 PLUS F2"
275 @" AND ";Q3(2);" PSIA O2 PLUS F2 PLUS HE"
280 @
285 @"BARRICADE I FILLED ON ";Y1$;" BARRICADE II FILLED ON ";Y2$;" OK ?"
290 INPUT D$
295 IF D$="o"THEN 860
300 IF D$="y"THEN 320
305 INPUT"B-I FILL DATE ",Y1$
```

```
310 INPUT"B-II FILL DATE ",Y2$
315 GOTO 285
320 @
325 @" B-I FILLED ON ";Y1$;" B-II FILLED ON ";Y2$
330 @
335 @"FOIL INSTALLED ";W$;" HAS BEEN USED ";N3;" TIMES, OK ?"
340 INPUT D$
345 IF D$="o"THEN 860
350 IF D$="y"THEN 370
355 INPUT"DATE",W$
360 INPUT"SHOTS ",N3
365 GOTO 335
370 @
375 @"FOIL INSTALLED ";W$;" HAS BEEN USED ";N3;" TIMES"
380 @
385 @"SHOT NO. = ";N1;" OK ?"
390 INPUT D$
395 IF D$="o"THEN 860
400 IF D$="y"THEN 415
405 INPUT"NEW SHOT NO. = ",N1
410 GOTO 385
415 @
420 @"SHOT NO. = ";N1
425 @
430 @"DIODE AREA = ";H;" BY ";L;" CM  , OK ?"
435 INPUT D$
440 IF D$="o"THEN 860
445 IF D$="y"THEN 465
450 INPUT"NEW HEIGTH ",H
455 INPUT"NEW WIDTH ",L
460 GOTO 430
465 @
470 @"DIODE AREA = ";H;" BY ";L;" CM"
475 @
480 @"MASTER SWITCH GAP = ";H1;" CM , OK ?"
485 INPUT D$
490 IF D$="o"THEN 860
495 IF D$="y"THEN 510
500 INPUT"NEW GAP ",H1
505 GOTO 480
510 @
515 @"MASTER SWITCH GAP = ";H1;" CM"
520 @
525 @"A-K GAP = ";H2;" CM , OK ?"
530 INPUT D$
535 IF D$="o"THEN 860
540 IF D$="y"THEN 555
545 INPUT"NEW GAP ",H2
550 GOTO 525
555 @
560 @"A-K GAP = ";H2;" CM"
565 @
570 @"DIODE PRESSURE = ";P4;" TORR , OK ?"
575 INPUT D$
580 IF D$="o"THEN 860
585 IF D$="y"THEN 600
590 INPUT"NEW DIODE PRESSURE ",P4
595 GOTO 570
600 @
605 @"DIODE PRESSURE = ";P4;" TORR"
610 @
615 @"CHARGING VOLTAGE = ";V1;" KV , OK ?"
620 INPUT D$
625 IF D$="o"THEN 860
630 IF D$="y"THEN 645
```

```
635 INPUT"NEW VOLTAGE ",V1
640 GOTO 615
645 @
650 @"CHARGING VOLTAGE = ";V1;" KV"
655 @
660 @"PULSE CHARGE SCALES - VERTICAL , HORIZ ";S1;" AND ";S2;" OK ?"
665 INPUT D$
670 IF D$="o"THEN 860
675 IF D$="y"THEN 695
680 INPUT"VERTICAL ",S1
685 INPUT"HORIZ ",S2
690 GOTO 660
695 @
700 @"VERTICAL = ";S1;" HORIZ = ";S2
705 @
710 @"TUBE VOLTAGE SCALES - VERTICAL , HORIZ ";S3;" AND ";S4;" OK ?"
715 INPUT D$
720 IF D$="o"THEN 860
725 IF D$="y"THEN 745
730 INPUT"VERTICAL ",S3
735 INPUT"HORIZ ",S4
740 GOTO 710
745 @
750 @"VERTICAL = ";S3;" HORIZ = ";S4
755 @
760 @"ROGOWSKI SCALES - VERTICAL , HORIZ ";S5;" AND ";S6;" OK ?"
765 INPUT D$
770 IF D$="o"THEN 860
775 IF D$="y"THEN 795
780 INPUT"VERTICAL ",S5
785 INPUT"HORIZ ",S6
790 GOTO 760
795 @
800 @"VERTICAL = ";S5;" HORIZ = ";S6
805 @
810 @"BARRICADES - VOLUME I , VOLUME II IN GALS ";V2(1);" AND ";V2(2)
815 INPUT D$
820 IF D$="o"THEN 860
825 IF D$="y"THEN 845
830 INPUT"VOLUME I ",V2(1)
835 INPUT"VOLUME II ",V2(2)
840 GOTO 810
845 @
850 @"VOLUME I = ";V2(1);" VOLUME II = ";V2(2)
855 @
860 M1=N1+1
865 M3=N3+1
870 ERASE"TEMP1"
875 ERASE"TEMP2"
880 ERASE"TEMP3"
885 CREATE"TEMP1"
890 OPEN¢1¢"TEMP1"
895 PUT¢1¢M1,H,L,H1,H2,P4,V1,R1,S1,S2,S3,S4,S5,S6,T,M3,R4,R5,R6
900 CREATE"TEMP2"
905 OPEN¢2¢"TEMP2"
910 PUT¢2¢V2(1),V2(2),Q1(1),Q1(2),Q2(1),Q2(2),Q3(1),Q3(2)
915 CREATE"TEMP3"
920 OPEN¢3¢"TEMP3"
925 PUT¢3¢Z$(-1),W$(-1),Y1$(-1),Y2$(-1)
930 CLOSE
935 READ C0,C1,C2,C3,C4,C5,C6,C7,C8
940 READ K5,K6,K7,K8,K9,X0(0),X1(0)
945 READ X2(0),X3(0),X4(0),X5(0),X6(0),X7(0),X8(0)
950 READ Y0(0),Y1(0),Y2(0),Y3(0),Y4(0)
955 DATA 6.35569,2.84127,-0.13327,-3.3,-0.135772,-1.04167,1,1,1
960 DATA 1,1,-0.40625,-0.68966,-0.666667,28.06565,0
965 DATA 0.68654,0,0,0,0,0,0
```

```
970 DATA 0,0,0.075,0.217,0.69983
975 SET 5,30000.0
980 @ : @ : @"PARAMETER SET-UP COMPLETE" : @
985 @"READY TO BEGIN TEST  ???"
990 INPUT D$
995 IF D$="y"THEN 1005
1000 GOTO 990
1005 T2(1)=0 : T2(2)=0
1010 C9=INP(%0014%) : GOSUB 5150
1015 X3(6)=C9 : C9=INP(%0017%) : GOSUB 5150
1020 X6(1)=C9 : C9=INP(%0017%) : GOSUB 5150
1025 X6(2)=C9 : I=0
1030 @
1035 ON ERROR GOTO 1040
1040 SET 5,10
1045 X3(5)=X3(6)
1050 C9=INP(%0014%) : GOSUB 5150
1055 X3(6)=C9
1060 C9=INP(%0017%) : GOSUB 5150
1065 X6(2)=C9 : C9=INP(%0013%) : GOSUB 5150
1070 X2=C9 : IF X2>-1 THEN 1080
1075 GOSUB 4000
1080 IF ABS(X6(2)-X6(1))<5 THEN 1090
1085 GOSUB 5000
1090 IF ABS(X3(6)-X3(5))>5 THEN 1110
1095 I=I+1
1100 IF FRA(I/30)=0 THEN 1030
1105 INPUT Z
1110 @I
1115 X3(1)=0 : X4(1)=0 : X5(1)=0
1120 X3(2)=0 : X4(2)=0 : X5(2)=0
1125 C9=INP(%0012%) : GOSUB 5150
1130 X1(2)=C9
1135 X1(6)=X1(2)
1140 @
1145 ON ERROR GOTO 1150
1150 SET 5,10
1155 C9=INP(%0014%) : GOSUB 5150
1160 X3=C9
1165 C9=INP(%0015%) : GOSUB 5150
1170 X4=C9
1175 C9=INP(%0016%) : GOSUB 5150
1180 X5=C9
1185 X1(1)=X1(2) : C9=INP(%0012%) : GOSUB 5150
1190 X1(2)=C9 : C9=INP(%0017%) : GOSUB 5150
1195 X6(2)=C9 : C9=INP(%0013%) : GOSUB 5150
1200 X2=C9 : IF X2>-1 THEN 1210
1205 GOSUB 4000
1210 IF ABS(X6(2)-X6(1))<5 THEN 1220
1215 GOSUB 5000
1220 IF ABS(X1(2)-X1(1))>2 THEN 1250
1225 T2(1)=T2(1)+1
1230 IF T2(1)<20 THEN 1245
1235 X3(1)=X3(1)+X3 : X4(1)=X4(1)+X4 : X5(1)=X5(1)+X5
1240 IF FRA(T2(1)/30)=0 THEN 1140
1245 INPUT Z
1250 @
1255 X1(7)=X1(1)
1260 X3(9)=X3(1)/(T2(1)-19) : X4(9)=X4(1)/(T2(1)-19)
1265 X5(9)=X5(1)/(T2(1)-19)
1270 D1=(X1(6)-X1(7))/T2(1)
1275 X1(8)=X1(2)
1280 Y2(2)=0 : Y3(2)=0 : Y4(2)=0
1285 @
1290 ON ERROR GOTO 1295
1295 SET 5,10
```

```
1300 C9=INP(%0011%) : GOSUB 5150
1305 X0=C9 : X1(1)=X1(2)
1310 C9=INP(%0012%) : GOSUB 5150
1315 X1(2)=C9 : C9=INP(%0014%) : GOSUB 5150
1320 X3=C9 : C9=INP(%0015%) : GOSUB 5150
1325 X4=C9 : C9=INP(%0016%) : GOSUB 5150
1330 X5=C9 : C9=INP(%0017%) : GOSUB 5150
1335 X6(2)=C9 : C9=INP(%001B%) : GOSUB 5150
1340 Y2=C9 : C9=INP(%001C%) : GOSUB 5150
1345 Y3=C9 : C9=INP(%001D%) : GOSUB 5150
1350 Y4=C9 : C9=INP(%001E%) : GOSUB 5150
1355 X7=C9 : C9=INP(%001F%) : GOSUB 5150
1360 X8=C9
1365 U=BINXOR(INP(%00D4%),%00FF%)
1370 IF U=128 THEN 1440
1375 C9=INP(%0013%) : GOSUB 5150
1380 X2=C9 : IF X2>-1 THEN 1390
1385 GOSUB 4000
1390 IF ABS(X6(2)-X6(1))<5 THEN 1405
1395 GOSUB 5000
1405 T2(2)=T2(2)+1
1410 IF T2(2)<20 THEN 1435
1415 X3(2)=X3(2)+X3 : X4(2)=X4(2)+X4 : X5(2)=X5(2)+X5
1420 Y2(2)=Y2(2)+Y2 : Y3(2)=Y3(2)+Y3 : Y4(2)=Y4(2)+Y4
1425 X7(2)=X7 : X8(2)=X8
1430 IF FRA(T2(2)/30)=0 THEN 1285
1435 INPUT Z
1440 @ : X1(9)=X1(1)
1445 X3(8)=X3(2)/(T2(2)-19) : X4(8)=X4(2)/(T2(2)-19)
1450 X5(8)=X5(2)/(T2(2)-19) : Y2(8)=Y2(2)/(T2(2)-19)
1455 Y3(8)=Y3(2)/(T2(2)-19) : Y4(8)=Y4(2)/(T2(2)-19)
1460 D2=(X1(8)-X1(9))/T2(2)
1465 DIM E4(120),E9(120)
1470 E4(0)=C7*X7(2) : E9(0)=C8*X8(2) : I=0
1475 @
1480 ON ERROR GOTO 1485
1485 SET 5,5
1490 I=I+1
1495 C9=INP(%001E%) : GOSUB 5150
1500 E4(I)=C7*C9
1505 C9=INP(%001F%) : GOSUB 5150
1510 E9(I)=C8*C9
1515 IF I=120 THEN 1530
1520 IF FRA(I/30)=0 THEN 1475
1525 INPUT Z
1530 @ : SET 5,0
1535 P8=C0*X0+X0(0) : P5(1)=C1*X1(6) : P6(1)=C1*X1(7)
1540 P5(2)=C1*X1(8) : P6(2)=C1*X1(9)
1545 P1=K8*Y3(8)+Y3(0) : P2=K7*Y2(8)+Y2(0) : P3=K9*Y4(8)+Y4(0)
1550 R4=C3*(X3(9)+X3(8))/2+X3(0)
1555 R5=C4*(X4(9)+X4(8))/2+X4(0)
1560 R6=C5*(X5(9)+X5(8))/2+X5(0)
1565 V3(1)=3.785*V2(1) : V3(2)=3.785*V2(2) : Q7(1)=51.7*Q3(1) : Q7(2)=51.7*Q3(2)
1570 Q4(1)=100*Q1(1)/Q7(1) : Q4(2)=100*Q1(2)/Q7(2)
1575 Q5(1)=100*(Q2(1)-Q1(1))/Q7(1) : Q5(2)=100*(Q2(2)-Q1(2))/Q7(2)
1580 Q6(1)=100.0-Q4(1)-Q5(1) : Q6(2)=100.0-Q4(2)-Q5(2)
1585 P7(1)=P5(1)-P6(1) : P7(2)=P5(2)-P6(2)
1590 R2(1)=P7(1)/T2(1) : R2(2)=P7(2)/T2(2)
1595 R3(1)=68.027*V3(1)*R2(1) : R3(2)=68.027*V3(2)*R2(2)
1600 N2(1)=INT((P6(1)-30)/P7(1)) : N2(2)=INT((P6(2)-30)/P7(2))
1605 L1=C3*X3(9)*T2(1) : L2=C3*X3(8)*T2(2)
1610 J1=C5*(X5(9)*T2(1)+X5(8)*T2(2)) : J4=C4*(X4(9)*T2(1)+X4(8)*T2(2))
1615 J2=L1*(1.0-14.7/760.0*Q2(1)/Q3(1))+L2*(1.0-14.7/760.0*Q2(2)/Q3(2))
1620 J3=14.7/760.0*(L1*(Q2(1)-Q1(1))/Q3(1)+L2*(Q2(2)-Q1(2))/Q3(2))
1625 J5=14.7/760.0*(L1*Q1(1)/Q3(1)+L2*Q1(2)/Q3(2))
1630 J6=J1+J2+J3+J4+J5
```

```
1635 A1=100*J1/J6 : A2=100*J2/J6 : A3=100*J3/J6 : A4=100*J4/J6 : A5=100*J5/J6
1640 P9=P8/76000.0
1645 B1=A1*P9
1650 B2=A2*P9 : B3=A3*P9 : B4=A4*P9 : B5=A5*P9
1655 C1=B1*760.0
1660 C2=B2*760.0 : C3=B3*760.0 : C4=B4*760.0 : C5=B5*760.0
1665 R1=1000000.0 : @ : @ : @
1670 @ : @ : @ : @ : @ : @ : @ : @
1675 @"E-BEAM LASER";TAB(29);"SHOT NO. ";N1;TAB(59);Z$
1680 @"TEST TIME ";T
1685 @TAB(33);"PAGE 1" : @
1690 @"*******************************************************************" : @
1695 @"E-BEAM PARAMETERS"
1700 @" SWITCH PRESSURES";TAB(45);"DIODE AREA = ";H;" BY ";L;" CM"
1705 @" TRIGATRON = ";P1;" PSIA N2";TAB(45);"A-K GAP = ";H2;" CM"
1710 @" MARX = ";P2;" PSIA SF6";TAB(45);"DIODE PRESSURE = ";P4;" TORR"
1715 @" PREPULSE = ";P3;" PSIA SF6";TAB(45);"CHARGING VOLTAGE = ";V1;" KV"
1720 @" MASTER SWITCH GAP = ";H1;" CM"
1725 @" WATER RESISTIVITY = ";R1;" OHM-CM"
1730 @ : @TAB(37);"SCOPE RECORDS" : @
1735 @TAB(15);"PULSE CHARGE";TAB(30);"******************************"
1740 @TAB(30);"*";TAB(60);"*"
1745 @TAB(30);"*";TAB(60);"*"
1750 @TAB(15);"VERTICAL";TAB(30);"*";TAB(60);"*";"   HORIZ"
1755 @TAB(30);"*";TAB(60);"*"
1760 @TAB(30);"*";TAB(60);"*"
1765 @TAB(16);S1;" KV/DIV";TAB(30);"*";TAB(60);"* ";S2;" NS/DIV"
1770 @TAB(30);"*";TAB(60);"*"
1775 @TAB(30);"*";TAB(60);"*"
1780 @TAB(30);"******************************" : @ : @ : @
1785 @TAB(15);"TUBE VOLTAGE";TAB(30);"******************************"
1790 @TAB(30);"*";TAB(60);"*"
1795 @TAB(30);"*";TAB(60);"*"
1800 @TAB(15);"VERTICAL";TAB(30);"*";TAB(60);"*";"   HORIZ"
1805 @TAB(30);"*";TAB(60);"*"
1810 @TAB(30);"*";TAB(60);"*"
1815 @TAB(16);S3;" KV/DIV";TAB(30);"*";TAB(60);"* ";S4;" NS/DIV"
1820 @TAB(30);"*";TAB(60);"*"
1825 @TAB(30);"*";TAB(60);"*"
1830 @TAB(30);"******************************" : @ : @ : @
1835 @TAB(15);"ROGOWSKI";TAB(30);"******************************"
1840 @TAB(30);"*";TAB(60);"*"
1845 @TAB(30);"*";TAB(60);"*"
1850 @TAB(15);"VERTICAL";TAB(30);"*";TAB(60);"*";"   HORIZ"
1855 @TAB(30);"*";TAB(60);"*"
1860 @TAB(30);"*";TAB(60);"*"
1865 @TAB(16);S5;" KV/DIV";TAB(30);"*";TAB(60);"* ";S6;" NS/DIV"
1870 @TAB(30);"*";TAB(60);"*"
1875 @TAB(30);"*";TAB(60);"*"
1880 @TAB(30);"******************************"
1885 @ : @ : @ : @ : @ : @ : @ : @ : @ : @ : @ : @ : @ : @ : @ : @ : @
1890 @"E-BEAM LASER";TAB(29);"SHOT NO. ";N1;TAB(59);Z$
1895 @"TEST TIME ";T
1900 @TAB(33);"PAGE 2" : @
1905 @"*******************************************************************" : @
1910 @"FOIL INSTALLED ";W$;" SHOTS WITH FOIL ";N3 : @
1915 @"GAS HANDLING" : @
1920 @"FLUORINE BARRICADE I STORAGE VOLUME IN USE = ";V2(1);" GALS (";V3(1);" LITERS)"
1925 @"BARRICADE I FILLED - ";Y1$;" TO ";Q1(1);" TORR OXYGEN, ";Q2(1);" TORR OXYGEN"
1930 @" PLUS FLUORINE, AND ";Q3(1);" PSIA OXYGEN PLUS FLUORINE PLUS HELIUM"
1935 @"BARRICADE I COMPOSITION - ";Q4(1);" % O2 ";Q5(1);" % F2 ";Q6(1);" % HE"
1940 @"BARRICADE I PRESSURES -"
1945 @"BEGINNING = ";P5(1);" PSIA";TAB(30);"END = ";P6(1);" PSIA"
1950 @"PRESSURE DROP = ";P7(1);" PSI";TAB(30);"ELAPSED TIME = ";T2(1);" SECS"
1955 @"DP/DT = ";R2(1);" PSI/SEC";TAB(30);"F2 MIX FLOWRATE = ";R3(1);" CC/SEC"
1960 @"THERE ARE ";N2(1);" SHOTS LIKE THIS LEFT IN BARRICADE I" : @
```

```
1965 @"FLUORINE BARRICADE II STORAGE VOLUME IN USE = ";V2(2);" GALS (";V3(2);" LITERS)"
1970 @"BARRICADE II FILLED - ";Y2$;" TO ";Q1(2);" TORR OXYGEN, ";Q2(2);" TORR OXYGEN"
1975 @" PLUS FLUORINE, AND ";Q3(2);" PSIA OXYGEN PLUS FLUORINE PLUS HELIUM"
1980 @"BARRICADE II COMPOSITION - ";Q4(2);" % O2 ";Q5(2);" % F2 ";Q6(2);" % HE"
1985 @"BARRICADE II PRESSURES -"
1990 @"BEGINNING = ";P5(2);" PSIA";TAB(30);"END = ";P6(2);" PSIA"
1995 @"PRESSURE DROP = ";P7(2);" PSI";TAB(30);"ELAPSED TIME = ";T2(2);" SECS"
2000 @"DP/DT = ";R2(2);" PSI/SEC";TAB(30);"F2 MIX FLOWRATE = ";R3(2);" CC/SEC"
2005 @"THERE ARE ";N2(2);" SHOTS LIKE THIS LEFT IN BARRICADE II" : @
2010 @"FLOW RATES";TAB(30);"PRESSURES" : @
2015 @" F2 MIX = ";R4;" CC/SEC";TAB(30);"LASER CHAMBER = ";P8;" TORR"
2020 @" H2     = ";R5;" CC/SEC"
2025 @" AR     = ";R6;" CC/SEC" : @
2030 @"MIXTURE COMPOSITION" : @
2035 @" GAS","PERCENT","   ATM","  TORR" : @
2040 @" ARGON",A1,B1,C1
2045 @" HELIUM",A2,B2,C2
2050 @" FLUORINE",A3,B3,C3
2055 @" HYDROGEN",A4,B4,C4
2060 @" OXYGEN",A5,B5,C5
2065 @ : @ : @ : @ : @ : @ : @ : @ : @ : @ : @ : @ : @ : @ : @ : @ : @ : @ : @ : @ : @ : @ : @ : @ : @
2070 @"E-BEAM LASER";TAB(29);"SHOT NO. ";N1;TAB(59);Z$
2075 @"TEST TIME ";T
2080 @TAB(33);"PAGE 3" : @
2085 @"**************************************************************************" : @
2090 @ : @TAB(33);"LASER OUTPUT" : @ : @
2095 @"  TIME (SECS)  9 CM CAL. (JOULES)       TIME (SECS)  4 CM CAL. (JOULES)"
2100 @
2105  FOR I=0 TO 40
2110  @TAB(6);(0.5*I-0.5);TAB(21);E9(I);TAB(44);(1.5*I-1.5);TAB(59);E4(3*I)
2115  NEXT I
2120 @ : @ : @ : @ : @ : @ : @ : @ : @ : @ : @ : @ : @ : @ : @ : @
2125 END
4000 @"DUMP TANK PRESSURE HIGH"
4005 OUT%0074%,%00FF%
4010 STOP
4015 RETURN
5000 @"MIXER VALVE OVERTEMPERATURE"
5005 OUT%0074%,%00FF%
5010 STOP
5015 RETURN
5150 IF C9<128 THEN 5170
5160 C9=C9-256
5170 RETURN

>>
```

Table A-1. Nomenclature

P0	Dump tank pressure, psia
P1, N2	Trigatron pressure, psia
P2, SF6	Marx pressure, psia
P3, SF6	Prepulse pressure, psia
H1	Master switch gap, cm
R1	Water resistivity, ohm-cm
H	Diode height, cm
L	Diode length, cm
H2	Anode-cathode (A-K) gap, cm
P4	Diode pressure, torr
V1	Charging voltage, kV
K	Mixer valve temperature, °C
T1	Elapsed time, sec
T2(1)	Barricade I use time, sec
T2(2)	Barricade II use time, sec
R3(1), R3(2)	Fluorine mix flow rates, cc/sec
N2(1), N2(2)	Shots remaining, integer
P8	Laser chamber pressure, torr
R4(1), R4(2)	Fluorine mix flow rates, cc/sec
R5	Hydrogen flow rate, cc/sec
R6	Argon flow rate, cc/sec
V2(1), V2(2)	Fluorine mix storage volumes, gal
V3(1), V3(2)	Fluorine mix storage volumes, l
E9	9-cm calorimeter, j
E4	4-cm calorimeter, j
S1-S6	Scope sensitivities
N1	Shot number
Q1(1), Q1(2)	Barricade, O_2, torr
Q2(1), Q2(2)	Barricade, $O_2 + F_2$, torr
Q3(1), Q3(2)	Barricade, $O_2 + F_2$ + He, psia
N3	Shots with foil

Table A-2. Port Definitions

Function	Analog Input/Ports	Offset	Counts	Scale Factors
Laser Chamber Pressure	DB 10 Pin 4 Port 11 H	X0(0)	X 0	C0
Barricade	DB 10 Pin 5 Port 12 H	X1(0)	X 1	C1
Dump Tank	DB 10 Pin 6 Port 13 H	X2(0)	X 2	C2
F_2 Mix Flow	DB 10 Pin 7 Port 14 H	X3(0)	X 3	C3
H_2 Flow	DB 10 Pin 8 Port 15 H	X4(0)	X 4	C4
Ar Flow	DB 10 Pin 9 Port 16 H	X5(0)	X 5	C5
Analog Temperature	DB 10 Pin 10 Port 17 H	X6(0)	X 6	C6
Diluent	DB 18 Pin 4 Port 19 H	Y0(0)	Y 0	K5
R1 Pressure	DB 18 Pin 5 Port 1A H	Y1(0)	Y 1	K6
Marx Pressure	DB 18 Pin 6 Port 1B H	Y2(0)	Y 2	K7
Trigatron Pressure	DB 18 Pin 7 Port 1C H	Y3(0)	Y 3	K8
Prepulse Pressure	DB 18 Pin 8 Port 1D H	Y4(0)	Y 4	K9
4-cm Calorimeter	DB 18 Pin 9 Port 1E H	X7(0)	X7 (m)	C7
9-cm Calorimeter	DB 18 Pin 10 Port 1F H	X8(0)	X8 (m)	C8